
Hoofdstuk 6.

Gestructureerde Types.

Gestructureerde types zijn verzamelingen van waarden die zelf uit meerdere componenten opgebouwd zijn. Complexe getallen, waarvan elke waarde een reëel en een imaginair deel heeft, zijn hiervan een goed voorbeeld. Modula heeft drie voorafgedefiniëerde gestructureerde types, ARRAYs, SETs en RECORDs, en biedt aan de programmeur de mogelijkheid zijn eigen gestructureerde types te definiëren. In dit hoofdstuk zullen de voorafgedefinieerde types besproken worden. In het derde deel van deze cursus (voor tweede kandidatuur Burgerlijk Ingenieur) wordt uitgelegd hoe men zelf types kan bij definiëren.

6.1. Het ARRAY type.

6.1.1. Beschrijving.

Het ARRAY type laat toe tabellen met elementen die allen van hetzelfde type zijn op te bouwen. De tabellen kunnen een willekeurig aantal dimensies hebben. De elementen worden geïdentificeerd door middel van een of meerdere indices, die zelf tot een ordinaal type moeten behoren. De syntax van een ARRAY type declaratie en van een specifiek component van een ARRAY vindt men in fig 6.1.

Het index type kan gelijk welk ordinaal type zijn terwijl het component type gelijk welk, enkelvoudig of gestruktureerd type kan zijn. De “index expression” moet, vanzelfsprekend, een waarde hebben die tot het index type behoort.

6.1.2. Voorbeeld.

Als voorbeeld toont tabel 6.1 een reservatietabel voor een lokaal.

TYPE 



Day

=
[Monday, Tuesday, Wednesday, Thursday,






Friday, Saturday, Sunday];



Hour

=
[8..20];



Course
=
ARRAY [0..29] OF CHAR;


Schedule
=
ARRAY Day,Hour OF Course;

VAR


Room213
:
Schedule;

BEGIN


...


Room213[Tuesday,14] := "Computer Systems";


...

END Example;

Tabel. 6.1. Reservatie tabel.

Het type Course is een één dimensionale tabel met 30 lettertekens, terwijl het type Schedule een twee dimensionale tabel is met 7 * 13 elementen, elk van het type Course. In feite had men Schedule dus ook kunnen definiëren als een drie dimensionale tabel met 7 * 13 * 30 lettertekens, maar de hier gebruikte hiërarchische benadering is zeker duidelijker.

6.1.3. De zeef van Erathostenes.

Als tweede voorbeeld van het gebruik van ARRAYs kan men het algoritme van Erathostenes, om priemgetallen kleiner dan een gegeven waarde te vinden, bespreken.

Eerst wordt er een verzameling gebouwd die alle getallen tussen 2 en de gewenste bovenlimiet Max bevat. Daarna wordt deze verzameling van klein naar groot doorlopen en wordt, voor elk getal dat nog aanwezig is, al zijn veelvouden verwijderd. De verzameling wordt voorgesteld door de variabele Sieve, een ARRAY [2..Max] OF BOOLEAN, waarvan elk element de waarde TRUE aanneemt wanneer het overeenstemmende getal aanwezig is in de verzameling.

Het volledige programma ziet er uit zoals weergegeven in tabel 6.2.:

MODULE Eratho;

  FROM InOut IMPORT WriteLn,WriteCard,WriteString;

  CONST Max = 1000; (* max value of primes *)

  VAR   i,m : CARDINAL;

        Sieve : ARRAY[2..Max] OF BOOLEAN;

BEGIN

  WriteString("Primes less than ");

  WriteCard(Max,5); WriteString(" are:");

  WriteLn;WriteLn;

(* Sieve initialisation *)

  FOR i := 2 TO Max DO Sieve[i] := TRUE END;

(* Selection of primes *)

  i := 2;

  WHILE i <= Max DO
    IF Sieve[i]

    THEN

      (* Write prime number *)

        WriteCard(i,5);

      (* Remove multiples from sieve *)

        m := i * 2;

        WHILE m <= Max DO
          Sieve[m] := FALSE; m := m+i;

        END;

    END;

    i := i + 1;

  END

END Eratho.

Tabel. 6.2. De zeef van Eratosthenes.

6.2. Het SET type.

6.2.1. Beschrijving.

Het type SET laat toe op eenvoudige wijze het wiskundige begrip van verzamelingen van ordinale waarden voor te stellen. In Modula 2 is het maximum aantal elementen van een SET afhankelijk van de implementatie van de taal en, in oudere implementaties van de taal, beperkt tot het aantal bits in een woord van de computer. Deze beperking liet toe de aanwezigheid of de afwezigheid van een element in een verzameling voor te stellen door één bit en een volledige verzameling door één woord, wat de snelheid van bewerkingen op SETs ten goede kwam, maar hun praktisch nut fel beperkte. In standaard Modula 2 krijgen sets die intern door de bits van een woord voorgesteld zijn de naam PACKEDSET. De waarde van een variabele van het type SET wordt toegekend door rechts van de toekenningsoperator een SET constructor te gebruiken, waarin de waarden die aanwezig zijn in de SET opgesomd worden. De voornaamste syntax diagrammen die betrekking hebben tot het SET type vindt men in fig.6.2.

Fig. 6.2. Syntax van SETs

Volgende bewerkingen zijn gedefiniëerd voor het type SET:

- De unie: (a+b) bevat alle elementen die tot a of b behoren.

- De intersectie: (a*b) bevat alle elementen die tot a en b behoren.

- Het verschil: (a-b) bevat alle elementen die tot a en niet tot b behoren.

- Het symmetrische verschil: (a/b) bevat alle elementen die tot een verzameling en niet tot de andere behoren.

Hierbij komt ook nog de BOOLEAN operator IN die toelaat na te gaan of een element al of niet aanwezig is in een verzameling.

Tenslotte zijn er de procedures (zie hoofdstuk 7) INCL en EXCL die toelaten op eenvoudige wijze een element aan een verzameling toe te voegen of te verwijderen.

6.2.2. Een eesrte voorbeeld van het gebruik van het type SET.

In tabel 6.3. vindt men de relevante delen van een programma dat het type SET gebruikt om allerlei statistieken te maken over de tekens die in een tekst voorkomen. De tekens van de tekst worden opeenvolgend gelezen (niet getoond) in een variabele Ch waarmee verder gewerkt wordt om de statistieken op te stellen. Het type CharSet wordt gedefiniëerd als een verzameling die alle tekens van een alfabet kan bevatten. Er worden vier constante sets gedefiniëerd: UppercaseLetters, LowercaseLetters, Vowels en Consonants waarvan de bedoeling vrij evident zijn. Tenslotte definiëert men ook nog een variabele verzameling CharactersUsed die gebruikt wordt om de verzameling tekens, die in de tekst gebruikt worden, op te bouwen tijdens het lezen. Hiervoor vertrekt men van een lege verzameling waaraan men, na het lezen van elk teken, dat teken toevoegt. In tabel 6.3 gebruikt men hiervoor de Unie van twee verzamelingen:

CharactersUsed := CharactersUsed + CharSet{Ch}; 

maar gebruik maken van de INCL procedure zou eleganter zijn:

INCL(CharactersUsed, Ch);

TYPE CharSet = SET OF CHAR;

...

CONST

  UppercaseLetters = {'A'..'C'};

  LowercaseLetters = {'a'..'z'};

  Vowels = {'A','E','I','O','U','a','e','i','o','u'};

  Consonants = UppercaseLetters + LowercaseLetters – Vowels;

...

VAR

  CharactersUsed : CharSet;

  Ch : CHAR;

  NUCLetters, NLCLetters, NVowels, NConsonants : CARDINAL;

...

BEGIN

(* Initialisations *)

  NUCLetters := 0; NLCLetters := 0; NVowels := 0; NConsonants := 0;

  CharactersUsed := CharSet{};

...

(* Updating Statistics *)

  IF     Ch IN UppercaseLetters THEN NUCLetters := NUCLetters + 1

   ELSIF Ch IN LowercaseLetters THEN NLCLetters := NLCLetters + 1

  END;

  IF     Ch IN Vowels           THEN NVowels := NVowels + 1

   ELSIF Ch IN Consonants       THEN NConsonants := NConsonants + 1

  END;

  CharactersUsed := CharactersUsed + CharSet{Ch};

...

END ...

Tabel 6.3. Voorbeeld van gebruik van het type SET

6.2.3. De zeef van Erathostenes.

Teneinde het gebruik van SETs verder te illustreren wordt in figuur 6.4 de zeef van Erathostenes opnieuw getoond, maar met een SET ipv. een ARRAY OF BOOLEAN. De verschillen zijn aangeduid door middel van schuinschrift. Wanneer het aantal elementen dat toegelaten is in een verzameling klein is heeft deze versie van het Erathostenes programma uiteraard weinig zin.

MODULE SEratho;

  FROM InOut IMPORT WriteLn,WriteCard,WriteString;

  CONST Max = 1000; (* max value of primes *)

  TYPE  SieveType = SET OF [2..Max];

  VAR   i,m : CARDINAL;

        Sieve : SieveType;

BEGIN

  WriteString ("Primes less than ");

  WriteCard(Max,5); WriteString(" are:");

  WriteLn;WriteLn;

(* Sieve initialisation *)

  Sieve := SieveType{2..Max};

(* Selection of primes *)

  i := 2;

  WHILE i <= Max DO

    IF i IN Sieve

    THEN

      (* Write prime number *)

        WriteCard(i,5);

      (* Remove multiples from sieve *)

        m := i * 2;

        WHILE m <= Max DO
          EXCL(Sieve,m); m := m+i;

        END;

    END;

    i := i + 1;

  END

END SEratho.

Tabel.6.4. De zeef van Erathostenes met SETs.

6.3. Het RECORD type.

6.3.1. Beschrijving.

Een RECORD laat toe verschillende variabelen, van willekeurige types, te groeperen in een enkele variabele. De verschillende onderdelen, die men velden noemt, dragen elk een naam. Zij kunnen gereferenceerd worden door de naam van de variabele gevolgd door een punt en door de naam van het veld, zoals geïllustreerd in de syntax diagrammen van fig.6.3.

6.3.2. Voorbeeld.

Men zou een type Date en een variabele Today kunnen definiëren zoals in Tabel 6.5: Normaal werkt men hoofdzakelijk met de variabele Today, zonder meer. Enkel wanneer men de details van de datum, de dag, de maand of het jaar nodig heeft, selecteert men specifieke velden van de variabele.

Fig.6.3. Syntax van RECORDS.

TYPE


Date
=
RECORD




Day   : [1..31];





Month : [1..12];





Year  : [1900..2100]




END;

VAR


Today : Date;

BEGIN

...


Today.Day    := 13;


Today.Month  := 09;


Today.Year   := 1998;

...

END ...

Tabel.6.5. Voorbeeld van een Record.

6.3.3. Het WITH statement.

Daar het referenceren van velden van RECORDs nogal veel schrijfwerk kan kosten, vooral wanneer de velden van een RECORD zelf RECORDs zijn, heeft men het WITH statement ingevoerd. Hiermee kan men, voor een ganse blok instructies opgeven met welke RECORD moet gewerkt worden, zonder telkens de naam van de variabele te moeten herhalen. 

Met het WITH statement zouden de waardetoekenningen van het vorige voorbeeld kunnen herschreven worden zoals in Tabel 6.6:


WITH Today DO


Day    := 25;



Month  := 12;



Year   := 1992


END; 

Tabel 6.6. Het gebruik van With.

Buiten het verminderen van het schrijfwerk laat het WITH statement ook optimisatie van de gegenereerde code toe, vermits het gebruik van een WITH statement een aanduiding voor de compiler zal zijn dat een variabele verschillende malen gaat gereferenceerd worden en dat het dus aangewezen is het adres van die variabele in een register te plaatsen.

6.3.4. Variante RECORDs.

Soms wenst men in één zelfde RECORD type verschillende combinaties van variabelen te kunnen gebruiken omdat, in functie van de waarde van sommige variabelen andere gegevens zinvol of zinloos kunnen zijn. Als men bijvoorbeeld een RECORD gebruikt om de technische gegevens van een wagen te groeperen, is het duidelijk dat de organisatie van dat RECORD niet dezelfde zal kunnen zijn voor een elektrische wagen als voor een wagen met een ontploffingsmotor. Om dergelijke verschillen op te vangen kan men een soort van CASE statement opnemen in de definitie van het RECORD. Een specifiek veld, "tag" genoemd geeft aan welke variante van de RECORD definitie dient gebruikt te worden. Wanneer de waarde van de "tag" eenvoudig van de context kan afgeleid worden, (in het geval van wagens zou het type motor van de aard van de brandstof kunnen afgeleid worden) volstaat het het type van de "tag" te vermelden, zonder een expliciet veld voor de waarde van de "tag" in het RECORD te voorzien. Wanneer zo gewerkt wordt, moet men er uiteraard zorgvuldig op letten dat de RECORDS op correcte wijze gebruikt worden, zonder te kunnen rekenen op enige waarschuwing in geval van vergissing.

Tabel 6.7 toont een RECORD dat een wagen beschrijft:

TYPE

  Car = RECORD
          Weight : CARDINAL;

          NumberDoors : CARDINAL;

          FrameId,EngineId : ARRAY[0..19] OF CHAR;

          Fuel : (Gasoline,FuelOil,LPG,Electricity);

          CASE Engine : (Explosion,Electricity) OF
            Explosion :

              NbrCylinders : CARDINAL;

              VolCylinders : CARDINAL |

            Electricity :

              Supply : (AC,DC);

              Voltage : CARDINAL;

          END;

        END;

Tabel 6.7. Variant record met expliciete tag.

Daar de aard van de motor eenvoudig kan afgeleid worden van het type brandstof zou men het veld Engine kunnen overslaan, zoals weergegeven in Tabel 6.8:

TYPE

  Car = RECORD
          Weight : CARDINAL;

          NumberDoors : CARDINAL;

          FrameId,EngineId : ARRAY[0..19] OF CHAR;

          Fuel : (Gasoline,FuelOil,LPG,Electricity);

          CASE (Explosion,Electricity) OF
            Explosion :

              NbrCylinders : CARDINAL;

              VolCylinders : CARDINAL |

            Electricity :

              Supply : (AC,DC);

              Voltage : CARDINAL;

          END;

        END;

Table.6.8. Variant record met impliciete tag.

Het essentiële van de syntax van de definitie van variante records wordt weergegeven in fig.6.4.

Fig.6.4. Syntax van het Variant Record type.

6.4. Combinaties van gestructureerde types.

Het is vanzelfsprekend dat de onderdelen van gestructureerde types zelf gestructureerde types mogen zijn. Zo is het bv. mogelijk aan een Array met vaste afmetingen een variabele toe te voegen om aan te duiden hoeveel elementen van de Array werkelijk gebruikt worden. Een dergelijk voorbeeld vindt men in de data declaratie van tabel 6.9. waarin het type StudentenLijst een record is dat naast een Array met vaste afmetingen ook nog een variabele bevat die kan gebruikt worden om aan te geven hoeveel elementen van de array nu effectief gebruikt worden.

CONST MaxAantalStudenten = 100;

TYPE

  AantalStudentenType = [0 .. MaxAantalStudenten];

  StudentenFiche = RECORD
                     (* details zinloos voor dit voorbeeld *)

                   END (* StudentenFiche *);

  StudentenLijst = RECORD
                     AantalStudenten : AantalStudentenType

                     Studenten       : ARRAY AantalStudentenType OF
                                         StudentenFiche

                   END (* StudentenLijst *)

Tabel 6.9. Voorbeeld van tabel met inhoudsteller.

